

Forth Challenge September 26, 2020 - Brad Nelson

```

#! /usr/bin/env gforth
: cap?  dup 1 and 0= ;   : >cap  cap? 0= if 95 and then emit ;
: letters1 [char] z 1+ [char] a do i >cap loop cr ;
: letters2 dup 123 < if dup >cap 1+ recurse exit then drop cr ;
: letters3 [char] a [char] A 26 0 do i 1 and if over else dup then emit
  1+ swap 1+ swap loop 2drop ;
: letters  cr letters1 [char] a letters2 letters3 cr ;

: _  2dup 0= swap 0= and if bl hold else # then ;
: _s  begin _ 2dup or 0= until ;   : ____. ( n -- ) 0 <# _ _ _ _s #> type ;
: gcd ( a b -- n ) dup if swap over mod recurse else drop then ;
: head  cr 7 spaces 250 240 do i _____. loop cr cr ;
: row ( n -- ) 250 240 do dup i gcd _____. loop drop cr ;
: chart  head 180 170 do i . 3 spaces i row loop ;

: square ( n -- n ) dup * ;
: star  42 emit ;   : stars ( n -- ) 0 ?do star loop ;
: graph cr cr 11 1 do i _____. space i square dup 1+ 2/ stars space . cr loop cr ;

: prime?  dup 2 ?do dup i mod 0= if drop 0 unloop exit then loop drop -1 ;
: prime. ( n -- n ) begin 1- dup prime? until dup . ;
: 5primes  cr 10000 5 0 do prime. loop drop cr ;

: esc 27 emit ;   : c esc ." [38;5;" 0 <# #s #> type ." m" ;
: all  page 196 c letters 40 c chart 69 c graph 214 c 5primes ;  all bye

```

```

AbCdEfGhIjKlMnOpQrStUvWxYz
AbCdEfGhIjKlMnOpQrStUvWxYz
AbCdEfGhIjKlMnOpQrStUvWxYz

      240 241 242 243 244 245 246 247 248 249

170      10  1  2  1  2  5  2  1  2  1
171      3  1  1  9  1  1  3 19  1  3
172      4  1  2  1  4  1  2  1  4  1
173      1  1  1  1  1  1  1  1  1  1
174      6  1  2  3  2  1  6  1  2  3
175      5  1  1  1  1 35  1  1  1  1
176     16  1 22  1  4  1  2  1  8  1
177      3  1  1  3  1  1  3  1  1  3
178      2  1  2  1  2  1  2  1  2  1
179      1  1  1  1  1  1  1  1  1  1

1 * 1
2 ** 4
3 ***** 9
4 ***** 16
5 ***** 25
6 ***** 36
7 ***** 49
8 ***** 64
9 ***** 81
10 ***** 100

9973 9967 9949 9941 9931

```